Criando Labels em 3D
Veremos nesta dica como criar labels capazes de exibir textos em 3D. Os labels deste exemplo terão  dois efeitos possíveis: destacados do formulário, com um pequeno sombreamento por trás dele ('Raised')  ou 'mergulhados" para dentro do formulário ('Lowered'). O procedimento responsável por estes efeitos é muito simples e útil. Este procedimento poderá facilmente ser personalizado para aceitar mais parâmetros.
Para testá-lo, inicie uma nova aplicação e insira o código abaixo na seção Implementation:
procedure Create3DLabel (oForm : TWinControl; const sText : String = 'Delphi';
                                            iTop : Integer = 0; iLeft : Integer = 0;
                                            iDepth : Integer = 3; sFontName : String = 'Arial';
                                            iFontSize : Integer = 10;
                                            sLookingType : String = 'Raised';
                                            sCorLabel : TColor = clBlack;
                                            sCor3D : TColor = clWhite);
var
    oLabel1 : TLabel;
    oLabel2 : TLabel;
begin
    oLabel1 := TLabel.Create(oForm);
    oLabel1.Parent := oForm;
    oLabel1.Transparent := True;
    oLabel1.Font.Name := sFontName;
    oLabel1.Font.Size := iFontSize;
    oLabel1.Caption := sText;

    oLabel2 := TLabel.Create(oForm);
    oLabel2.Parent := oForm;
    oLabel2.Transparent := True;
    oLabel2.Font.Name := sFontName;
    oLabel2.Font.Size := iFontSize;
    oLabel2.Caption := sText;

    if sLookingType = 'Lowered' then
    begin
        oLabel1.Top := iTop + iDepth;
        oLabel1.Left := iLeft + iDepth;
        oLabel1.Font.Color := sCor3D;

        oLabel2.Font.Color := sCorLabel;
        oLabel2.Top := iTop;
        oLabel2.Left := iLeft;
        oLabel2.BringToFront;
    end;

    if sLookingType = 'Raised' then
    begin
        oLabel1.Top := iTop - iDepth;
        oLabel1.Left := iLeft - iDepth;
        oLabel1.Font.Color := sCorLabel;
        oLabel1.BringToFront;

        oLabel2.Top := iTop;
        oLabel2.Left := iLeft;
        oLabel2.Font.Color := sCor3D;
    end;
end;
Adicione dois botões ao formulário e no evento OnClick de cada um escreva o seu respectivo código como mostrado a seguir:
procedure TForm1.Button1Click(Sender: TObject);
var
    strCaption, strFonte : string;
    intTamanhoFonte, intPosEsquerda, intPosCima : integer;
    strCorLabel : TColor;
const
    str3DParaDentro : string = 'Lowered';
    strCorFundo : TColor = clWhite;
    intProfDentro : integer = 2;
begin
    strCaption := 'Hello World';
    strFonte := 'Times New Roman';
    intTamanhoFonte := 72;
    strCorLabel := clBlue;
    intPosEsquerda := 10;
    intPosCima := 10;
    Create3DLabel (TWinControl(self), strCaption, intPosCima, intPosEsquerda,
                                intProfDentro, strFonte, intTamanhoFonte, str3DParaDentro,
                                strCorLabel, strCorFundo);
end;

procedure TForm1.Button2Click(Sender: TObject);
var
    strCaption, strFonte : string;
    intTamanhoFonte, intPosEsquerda, intPosCima : integer;
    strCorLabel : TColor;
const
    str3DParaFora : string = 'Raised';
    strCorFundo : TColor = clBlack;
    intProfFora : integer = 3;
begin
    strCaption := 'Hello World';
    strFonte := 'Times New Roman';
    intTamanhoFonte := 72;
    strCorLabel := clRed;
    intPosEsquerda := 100;
    intPosCima := 100;
    Create3DLabel (TWinControl(self), strCaption, intPosCima, intPosEsquerda,
                                intProfFora, strFonte, intTamanhoFonte, str3DParaFora,
                                strCorLabel, strCorFundo);
end;
Implementando a Função SubString
Abordaremos nesta dica, a criação de uma função que retorne uma parte específica de uma string. A função, que chamaremos de SubString, deverá receber os seguintes parâmetros: 1o.) a string original; 2o.) qual parte se deseja retirar da string original, por exemplo, parte 1, parte 2, etc e 3o.) o separador que será utilizado para dividir a string original. Este separador deverá constar na string original, caso contrário o resultado será a string original e não uma parte dela. 
Exemplo: SubString('PRIMEIRO SEGUNDO TERCEIRO', 1, ' ') = 'SEGUNDO'
OBs.: 1) Na função SubString, o índice do parâmetro 'Parte' é base-zero, ou seja, Parte = 0 retornará a primeira substring, Parte = 1 retornará a segunda, e assim sucessivamente; 
         2) A função SubString é sensível ao texto, portanto certifique-se de digitar no parâmetro 'Separador', o caractere respeitando o formato da string original, caso contrário o resultado será a string original e não uma parte dela. 
         3) No exemplo acima o separador utilizado é um espaço em branco (#32). 

Etapa 1) Crie um novo projeto e insira no formulário os seguintes componentes da palheta Standard: 4 Edits, 4 Labels e 1 Botão. Monte o Layout do formulário para que fique igual o desenho abaixo. Altere também os Captions dos Labels da seguinte forma: 

         Label1 = String Original: 
         Label2 = Parte: 
         Label3 = Separador: 
         Label4 = Resultado: 


Etapa 2) Escreva o código a seguir, de forma que a sua Unit se pareça com o texto abaixo:

unit untSubString; 

interface 

uses 
    Windows, Messages, SysUtils, Variants, Classes, Graphics, Controls, Forms, 
    Dialogs, StdCtrls; 

type 
    TForm1 = class(TForm) 
        Button1: TButton; 
        edtStringOriginal: TEdit; 
        edtParte: TEdit; 
        edtSeparador: TEdit; 
        edtResultado: TEdit; 
        Label1: TLabel; 
        Label2: TLabel; 
        Label3: TLabel; 
        Label4: TLabel; 
        procedure Button1Click(Sender: TObject); 
    private 
        { Private declarations } 
    public 
        { public declarations } 
    end; 

var 
    Form1: TForm1; 

implementation 

{$R *.dfm} 

function SubString(Str : string; Parte : integer; Separador : string) : string; 
var 
    lp0, sln, CurrN : integer; 
    EmSep, PrimEncontrado : boolean; 
begin 
    result := ''; 
    CurrN := 0; 
    EmSep := false; 
    PrimEncontrado := false; 
    sln := length(Str); 
    for lp0 := 1 to sln do begin 
        if Pos(Str[lp0],Separador) > 0 then begin 
           if not EmSep then begin 
              if PrimEncontrado then inc(CurrN); 
           end; 
           EmSep := true; 
           if CurrN > Parte then break; 
        end else begin 
           EmSep := false; 
           PrimEncontrado := true; 
           if CurrN = Parte then result := result + Str[lp0]; 
    end; 
end; 
end; 

procedure TForm1.Button1Click(Sender: TObject); 
var 
    strOriginal, strSeparador, strResultado : String; 
    intParte : Integer; 
begin 
    strOriginal := edtStringOriginal.Text; 
    intParte := StrToInt(edtParte.Text); 
    strSeparador := edtSeparador.Text; 
    strResultado := SubString(strOriginal, intParte, strSeparador); 
    edtResultado.Text := strResultado 
end; 

end. 
